



Fundamentos de la Web Semántica

Lenguaje de consulta SPARQL


Arquitectura de la Web Semántica




Arquitectura de la Web Semántica



Arquitectura de la Web Semántica



Arquitectura de la Web Semántica













Por qué se necesita un lenguaje de consulta de grafos?

Buscamos en Google “bailarinas y bailarines de ballet”

bailarinas y bailarines de ballet


Todo Imágenes Videos Noticias Maps Más Preferencias Herramientas

Bailarinas / ballet / Bailarín

									
Rudolf Nureyev 1938–1993	Bob Fosse 1927–1987	Roland Petit 1924–2011	George Balanchine 1904–1983	Vaslav Nijinsky 1889–1950	Misty Copeland	Anna Pávlova 1881–1931	Mijail Baryshnikov	Maya Plisetskaya 1925–2015	Marian Núñez

Imágenes de bailarinas y bailarines de ballet

peso estatura lesiones danza clásica ballet clásico tutu ballerina



Notificar imágenes

→ Ver todos

Videos

TOP 10 8:05

Top 10 Las Mejores bailarinas de ballet en el mundo / the best ...

YouTube · perfecta de pies a cabeza
16 jun. 2016

Por qué se necesita un lenguaje de consulta de grafos?

Obtenemos parte de la información que buscamos

RDF y OWL **no son suficientes** si queremos obtener un conjunto de recursos que cumplen determinadas condiciones.

Bases RDF ayudan en la búsqueda: describen recursos usando vocabularios estándares

El **lenguaje SPARQL** nos permite **obtener directamente un subgrafo con los datos que buscamos**, usando dichos vocabularios

Por qué se necesita un lenguaje de consulta de grafos?

“Bailarinas y bailarines de ballet”

```
select distinct ?bailarina ?nombre
where
{
  ?bailarina rdf:type dbo:Artist.
  OPTIONAL {?bailarina dbo:birthName ?nombre. }
  {
    {?bailarina dbo:field dbr:Ballet} UNION
    {?bailarina rdf:type yago:BalletDancer109834699} UNION
    {?bailarina rdf:type yago:Ballerina109834592}
  }
}
```



Por qué se necesita un lenguaje de consulta de grafos?

“Bailarinas y bailarines de ballet”

bailarina	nombre
http://dbpedia.org/resource/Jeffrey_Golladay	"Jeffrey Golladay"@en
http://dbpedia.org/resource/Arron_Scott	"Arron Scott"@en
http://dbpedia.org/resource/Zaira_Cosico	"Zaira Cosico"@en
http://dbpedia.org/resource/Keith_McMillan	"Keith Lionel McMillan"@en
http://dbpedia.org/resource/Lisa_Macuja-Elizalde	"Lisa Teresita Pacheco Macuja"@en
http://dbpedia.org/resource/Igor_Yebra	"Igor Yebra Iglesias"@en
http://dbpedia.org/resource/Jennifer_Olayvar	"Jennifer Rose Olayvar"@en
http://dbpedia.org/resource/Georgette_Sanchez	"Georgina Johanna Garcia Sanchez"@en
http://dbpedia.org/resource/María_Noel_Ricetto	"Maria Noel Ricetto"@en
http://dbpedia.org/resource/Yvonne_Chouteau	"Myra Yvonne Chouteau"@en
http://dbpedia.org/resource/Luciana_Savignano	
http://dbpedia.org/resource/Michele_Wiles	
http://dbpedia.org/resource/Carolyn_George	
http://dbpedia.org/resource/Christopher_Wheeldon	
http://dbpedia.org/resource/Christopher_d'Amboise	
http://dbpedia.org/resource/Gemma_Bond	
http://dbpedia.org/resource/Sergei_Vikharev	
http://dbpedia.org/resource/Peter_Quanz	
http://dbpedia.org/resource/Esteban_Berlanga	"Esteban Berlanga"@en
http://dbpedia.org/resource/Pyotr_Gusev	"Pyotr Andreyevich Gusev"@en
http://dbpedia.org/resource/Dominic_Antonucci	"Dominic Antonucci"@en
http://dbpedia.org/resource/Jane_Gurnett	"Jane E. Gurnett"@en
http://dbpedia.org/resource/Sabrina_Matthews	"Sabrina Matthews"@en
http://dbpedia.org/resource/Adabel_Guerrero	"Adabel Anahí Guerrero Melachenko"@en
http://dbpedia.org/resource/Viper_(actress)	"Stephanie Green"@en
http://dbpedia.org/resource/Gregori_Lukas	
http://dbpedia.org/resource/Oh_Land	

Por qué se necesita un lenguaje de consulta de grafos?

“Bailarinas y bailarines de ballet”



Lenguaje de consulta SPARQL

Estándar W3C desde 2008:


- Sintaxis y semántica del lenguaje
- Protocolo de transmisión de consultas y resultados hacia y desde las bases RDF que ofrecen **endpoints**

Extensión del lenguaje (1.1): subconsultas, agregaciones, negación

Consultas federadas: formular **consultas distribuidas integrando datos de diferentes bases RDF.**

Endpoints: GraphDB, DBPEDIA, WIKIDATA...

DBpedia



Patrón de la consulta

Conjunción de triplas donde sujeto, predicado y objeto pueden ser una variable.



Obtener el lugar de nacimiento de María Noel Riccetto y país al que pertenece:

```
{  
  dbr:María_Noel_Riccetto dbo:birthPlace ?lugarNac .  
  ?lugarNac dbo:country ?país .  
}
```

Variables

Las flechas rojas indican que el texto 'Variables' se refiere a los marcadores de interrogación '?lugarNac' y '?país' en la consulta SQL anterior.

Patrón de la consulta

Formalmente:

I : conjunto de IRIs
 B : conjunto de nodos blancos
 L : conjunto de literales
 V : conjunto de variables

} disjuntos

Triplas $(s, p, o) \in (I \cup B) \times I \times (I \cup B \cup L)$

Triple pattern $t \in (I \cup B \cup V) \times (I \cup V) \times (I \cup B \cup L \cup V)$

Patrón de consulta básico (Basic graph pattern):

$BGP = \{t \in (I \cup B \cup V) \times (I \cup V) \times (I \cup B \cup L \cup V)\}$

Consultas SPARQL

Se basan en un **patrón de consulta**

Recuperan los recursos que cumplen con las **condiciones** del patrón de consulta.

Patrón de consulta básico: se corresponde con un **subgrafo** de datos RDF (**resultado**) cuando los recursos del subgrafo pueden ser sustituidos en las variables.

El resultado puede ser serializado en diferentes formatos.

Tipos de consultas SPARQL

SELECT: conjunto de datos RDF que coinciden con un patrón de consulta.

CONSTRUCT: nuevo grafo RDF a partir de datos RDF que coinciden con un patrón de consulta.

ASK: si existen datos RDF coincidentes con un patrón de consulta.

DESCRIBE: grafo RDF que describe los recursos que cumplen con un patrón de consulta.

SPARQL - SELECT

Sintaxis general de las consultas:

Prefijos

SELECT.... Descripción del resultado: qué información se quiere obtener

FROM.... Grafo(s) RDF: dónde buscar

WHERE..... Patrón de la consulta: qué buscar

ORDER BY.... Modificadores de la consulta: filtrar y organizar los resultados

SELECT, WHERE: obligatorios

SPARQL - SELECT

Sintaxis general de las consultas:

Prefijos

SELECT.... Descripción del resultado: qué información se quiere obtener

FROM.... Grafo(s) RDF: dónde buscar

WHERE.... Patrón de la consulta: qué buscar

ORDER BY.... Modificadores de la consulta: filtrar y organizar los resultados

SELECT, WHERE: obligatorios

SPARQL vs SQL?

Similitud en la sintaxis pero: trabajan en **diferentes estructuras de datos**

SPARQL - SELECT

Endpoint: <https://dbpedia.org/sparql>.

```
select ?lugarNac ?pais
where
{
    dbr:María_Noel_Ricetto dbo:birthPlace ?lugarNac .
    ?lugarNac dbo:country ?pais .
}
```



SPARQL – Patrones complejos

Patrón de consulta básico: el resultado de la consulta debe satisfacer TODAS las condiciones.

La información en la web está incompleta

A veces es necesario recuperar un conjunto de recursos aunque no todas las propiedades tengan un valor asociado.

Patrón de consulta complejo : permite introducir variantes....

- Traer valores SI existen
- Recuperar un valor usando diferentes recursos

SPARQL – Patrones complejos

Patrón de consulta básico: el resultado de la consulta debe satisfacer TODAS las condiciones.

La información en la web está incompleta

A veces es necesario recuperar un conjunto de recursos aunque no todas las propiedades tengan un valor asociado.


Patrón de consulta complejo : permite introducir variantes....

- Traer valores SI existen → **OPTIONAL**
- Recuperar un valor usando diferentes recursos → **UNION**

SPARQL – Optional pattern

No siempre se puede asumir estructuras completas en los grafos RDF.

- La información puede ser agregada cuando esté disponible, por lo que es bueno no rechazar una solución cuando no coincide alguna parte del graph pattern:
- **OPTIONAL** matching: **no se asignan valores a las variables**, pero **no se elimina la solución**



Recuperar todos los recursos con lugar de nacimiento Montevideo, que tengan fecha de nacimiento y en caso que tengan, su página web

```
SELECT ?x ?y ?z
WHERE
{
  ?x dbo:birthPlace dbr:Montevideo.
  ?x dbo:birthDate ?y.
  OPTIONAL {?x foaf:homepage ?z.}
}
```



Recuperar todos los recursos con lugar de nacimiento Montevideo, que tengan fecha de nacimiento y en caso que tengan, su página web

```
SELECT ?x ?y ?z
WHERE
{
  ?x dbo:birthPlace dbr:Montevideo.
  ?x dbo:birthDate ?y.
  OPTIONAL {?x foaf:homepage ?z.}
}
```

```
SELECT ?x ?y ?z ?t
WHERE
{
  ?x dbo:birthPlace dbr:Montevideo.
  ?x dbo:birthDate ?y.
  OPTIONAL {?x foaf:homepage ?z.}
  OPTIONAL {?x dc:description ?t.}
}
```


Múltiples OPTIONAL
son validados en forma
INDEPENDIENTE

Recuperar todos los recursos con lugar de nacimiento en Uruguay, o en un departamento de Uruguay



Recuperar todos los recursos con lugar de nacimiento en Uruguay, o en un departamento de Uruguay



```
SELECT ?x ?y
WHERE
{
  {?x dbo:birthPlace ?y.
   ?y dbo:country dbr:Uruguay.}
UNION
 {?x dbo:birthPlace dbr:Uruguay.}
}
```



UNION: Permite combinar graph patterns → **OR NO EXCLUSIVO**


Resultado: unión de los resultados de todos los graph patterns

Ejercicio: Recuperar todos los recursos con lugar de nacimiento en Uruguay, o en un departamento de Uruguay, y en caso que tengan, la página web



Recuperar todos los recursos con lugar de nacimiento en Uruguay, o en un departamento de Uruguay, y en caso que tengan, la página web

```
SELECT ?x ?y ?z
WHERE
{
  {?x dbo:birthPlace ?y.
   ?y dbo:country dbr:Uruguay.}
  UNION
  {?x dbo:birthPlace dbr:Uruguay.}
  OPTIONAL {?x foaf:homepage ?z.}
}
```



SPARQL – Filtrado de soluciones

Para restringir la soluciones: **FILTER**


Recuperar las ciudades del Uruguay con más de 20.000 habitantes

SPARQL – Filtrado de soluciones

Para restringir la soluciones: **FILTER**

Recuperar las ciudades del Uruguay con más de 20.000 habitantes

```
SELECT ?x ?population
WHERE
{
  ?x  rdf:type  dbo:City.
  ?x  dbo:country  dbr:Uruguay.
  ?x  dbo:populationTotal  ?population.
  FILTER (?population > 20000)
}
```



Condición
adicional que
RESTRINGE el
resultado de la
consulta

SPARQL – Filtrado de soluciones

Filters:

- No están basados estrictamente en el modelo de datos RDF (como los patrones de consulta)
- **FILTER** (?population < 100) → evalúan a “true” o “false”
- Resultado de la consulta: además de **satisfacer el patrón de la consulta** debe evaluar a “true” la **expression del filter**
- Aplica al **grupo** del patrón de la consulta **dentro del que aparece el FILTER**

SPARQL – Filtrado de soluciones

Expresiones del FILTER: funciones

- Operadores de **comparación**: < = > <= >= !=

Todos para xsd:boolean, xsd:string, xsd:dateTime, numéricos (como xsd:integer) y literales no tipados (excluyendo language settings).

Para otros tipos: solo =, !=

- Operadores **especiales**: isURI(A), isBLANK(A), isLITERAL(A), sameTERM(A, B), REGEX(A, B).....
- Operadores **booleanos**, para combinar condiciones: && || !
- Operaciones **aritméticas**: + - * /

Combinando FILTER y OPTIONAL

Ejercicio: Recuperar todas las ciudades de Uruguay y, para las que tienen entre de 20.000 y 500.000 habitantes recuperar su población

Combinando FILTER y OPTIONAL

Recuperar todas las ciudades de Uruguay y, para las que tienen entre de 20.000 y 500.000 habitantes recuperar su población

```
SELECT ?x ?population
WHERE
{
  ?x rdf:type dbo:City.
  ?x dbo:country dbr:Uruguay.
  OPTIONAL {?x dbo:populationTotal ?population.
    FILTER (?population > 20000 && ?population < 500000)}
}
```

SPARQL – Solution modifiers

Query patterns generan un conjunto de soluciones en cualquier orden, pero a veces es necesario reorganizar el resultado.

- Cómo hacemos para recuperar solamente **parte** de los resultados?
- Cómo **ordenamos** los resultados?
- Cómo **eliminamos** resultados **repetidos**?

SPARQL – Solution modifiers

ORDER BY: ordenar las soluciones

Recuperar todas las ciudades de Uruguay y, para las que tienen entre de 20.000 y 500.000 , recuperar su población. Ordenarlas por su nombre.

```
SELECT ?x ?name ?population
WHERE
{
  ?x rdf:type dbo:City.
  ?x foaf:name ?name.
  ?x dbo:country dbr:Uruguay.
  OPTIONAL {?x dbo:populationTotal ?population.
             FILTER (?population > 20000 && ?population < 500000)}
}
ORDER BY ?name
```

SPARQL – Solution modifiers

ORDER BY: ordenar las soluciones

Recuperar todas las ciudades de Uruguay y, para las que tienen menos de 100.000, recuperar su población. Ordenarlas por su nombre.

```
SELECT ?x  ?name ?population
WHERE
{
  ?x  rdf:type  dbo:City.
  ?x  foaf:name ?name.
  ?x  dbo:country dbr:Uruguay.
  OPTIONAL {?x  dbo:populationTotal ?population.
            FILTER (?population > 20000 && ?population < 500000)}
}
ORDER BY ASC(?population), DESC(?name)
```

Orden alfabético de URIs como strings

Orden entre elementos de diferentes tipos:

unbound variables < nodos blancos < URIs < literales RDF

SPARQL – Solution modifiers

DISTINCT: eliminar soluciones repetidas

LIMIT: máxima cantidad de resultados

OFFSET: posición del primer resultado

Orden de aplicación de los modificadores:

- **ORDER BY**
- Eliminar duplicados (**DISTINCT**)
- Eliminar resultados iniciales (**OFFSET**)
- Eliminar resultados posteriores a la cantidad dada por **LIMIT**

SPARQL – Construcción de grafos

CONSTRUCT: permite construir grafos

Se generan triplas RDF usando los resultados que devuelve el graph pattern, que se sustituyen en un **graph template**.

Prefix `dcterm`s → <http://purl.org/dc/terms/>

CONSTRUCT

```
{ ?place foaf:name ?name }
```

WHERE

```
{  
  ?place      dbo:country   dbr:Uruguay.  
  ?place      dcterm:subject ?name.  
}
```


SPARQL – Construcción de grafos

CONSTRUCT: permite construir grafos

Se generan triplas RDF usando los resultados que devuelve el graph pattern, que se sustituyen en un **graph template**.

Prefix **dct** → <http://purl.org/dc/terms/>

```
CONSTRUCT                                graph template
{ ?place foaf:name ?name }
WHERE
{
  ?place    dbo:country    dbr:Uruguay.
  ?place    dcterms:subject ?name.
}
```

SPARQL – ASK

SPARQL permite validar si un query pattern tiene alguna solución

ASK{

```
?place    dbo:country    dbr:Uruguay.
```

```
?place    dbo:populationTotal    ?population.
```

```
FILTER (?population < 1000)
```

```
}
```

SPARQL – DESCRIBE

DESCRIBE: devuelve un grafo RDF que describe el recurso devuelto por el query pattern.

```
DESCRIBE ?place
WHERE
{
  ?place    foaf:name    "Montevideo"@en.
}
```

SPARQL 1.1 – Extensiones del lenguaje

- **Expresiones en SELECT, BIND - VALUES**
- **Agregaciones**
- **Subconsultas**
- **Property paths**
- **Negación**

SPARQL 1.1 - Expresiones en SELECT, BIND

Recuperar el nombre y la población de los lugares de Uruguay, expresada en miles.

```
SELECT  ?place ?name (?pop/1000) AS ?population
WHERE
{
  ?place    dbo:country    dbr:Uruguay.
  ?place    dbo:populationTotal  ?pop.
  ?place    foaf:name      ?name
}
ORDER BY ?population
```

```
SELECT  ?place ?name ?population
WHERE
{
  ?place    dbo:country    dbr:Uruguay.
  ?place    dbo:populationTotal  ?pop.
  ?place    foaf:name      ?name
}
BIND (?pop/1000 AS ?population)
ORDER BY ?population
```

SPARQL 1.1 - VALUES

Obtener el nombre y la población de la ciudad de Montevideo , expresada en miles.

```
SELECT  ?name (?pop/1000) AS ?population
WHERE
{
  ?place  dbo:populationUrban  ?pop.
  ?place  foaf:name  ?name .
  VALUES ?place {dbr:Montevideo}
}
```

SPARQL 1.1 - Agregaciones

Permiten agrupar soluciones y hacer cálculos sobre los grupos

Recuperar la población de los lugares de Uruguay agrupada por departamento.

```
SELECT  ?depto (SUM(?pop) as ?popdepto)
WHERE
{
  ?place dbo:country    dbr:Uruguay.
  ?place  dbo:populationTotal  ?pop.
  ?place  dbo:subdivision  ?depto.
}
GROUP BY ?depto
```

SPARQL 1.1 - Agregaciones

Permiten agrupar soluciones y hacer cálculos sobre los grupos

Recuperar la población de los lugares de Uruguay agrupada por departamento , para aquellos departamentos con más de 200000 habitantes..

```
SELECT  ?depto (SUM(?pop) as ?popdepto)
WHERE
{
  ?place dbo:country    dbr:Uruguay.
  ?place  dbo:populationTotal  ?pop.
  ?place  dbo:subdivision  ?depto.
}
```

GROUP BY ?depto

HAVING (SUM(?pop) > 200000)

→ Agrupar las soluciones

→ Filtrar valores agregados

SUM, COUNT, MIN, MAX, AVG, GROUP_CONCAT, SAMPLE → Funciones de agregación

SPARQL 1.1 - Subconsultas

Recuperar los departamentos de Uruguay con más de 200.000 habitantes.

```
SELECT ?depto ?nombre
WHERE
{
  {?depto rdfs:label ?nombre.
  {
    SELECT ?depto (SUM(?pop) as ?popdepto)
    WHERE
      {
        ?place dbo:country dbr:Uruguay.
        ?place dbo:populationTotal ?pop.
        ?place dbo:subdivision ?depto.
      }
    GROUP BY ?depto
    HAVING (SUM(?pop) > 200000)
  }
}
```

SPARQL 1.1 - Property paths

Recuperar el nombre de la capital de Uruguay y el nombre de su fundador.

```
SELECT ?nombreCap ?nombreFund
WHERE
{
  dbr:Uruguay dbo:capital ?capital.
  ?capital foaf:name ?nombreCap.
  ?capital dbo:founder ?fundador.
  ?fundador rdfs:label ?nombreFund.
}
```

SPARQL 1.1 - Property paths

Recuperar el nombre de la capital de Uruguay y el nombre de su fundador.

```
SELECT ?nombreCap ?nombreFund
WHERE
{
  dbr:Uruguay dbo:capital ?capital.
  ?capital foaf:name ?nombreCap.
  ?capital dbo:founder ?fundador.
  ?fundador rdfs:label ?nombreFund.
}
```

```
SELECT ?nombreCap ?nombreFund
WHERE
{
  dbr:Uruguay dbo:capital ?capital.
  ?capital foaf:name ?nombreCap.
  ?capital dbo:founder/rdfs:label ?nombreFund.
}
```

Expresiones regulares
en los predicados

SPARQL 1.1 - Property paths

Otras expresiones regulares:

- Caminos de largo arbitrario: `?s ex:p+ ?o`, `?s ex:p* ?o`
- Caminos alternativos: `?s (ex:p1|ex:p2) ?o`
- Negación de caminos: `?s !ex:p ?o`
- Caminos inversos: `?s ^ex:p ?o` es lo mismo que `?o ex:p ?s`
- Secuencia de caminos: `?s ex:p1 / ex:p2 ?o`
- Camino de largo 0 ó 1: `?s ex:p? ?o`

SPARQL 1.1 - Negación

Dos formas de negación:

- Eliminar soluciones que **cumplen con un graph pattern**

```
SELECT ?a ?b
```

```
WHERE
```

```
{?a dbo:birthDate ?b.
```

```
  FILTER NOT EXISTS {?a dbo:birthName ?c}}
```

```
SELECT ?a ?b
```

```
WHERE
```

```
{?a dbo:birthDate ?b.
```

```
  MINUS {?a dbo:birthName ?c}}
```

SPARQL 1.1 - UPDATE

Manipulación de grafos y su contenido

- Adición y eliminación de grafos
- Adición y eliminación de triplas en un grafo
- LOAD, DROP, CREATE
- INSERT, DELETE para triplas
- No hay transacciones

```
DELETE { ?person foaf:givenName "Bill" }  
INSERT { ?person foaf:givenName "William" }  
WHERE {  
        ?person a foaf:Person .  
        ?person foaf:givenName "Bill"  
}
```

Grafos nombrados

```
select ?lugarNac ?pais
where
{
    dbr:María_Noel_Riccetto dbo:birthPlace ?lugarNac .
    ?lugarNac dbo:country ?pais .
}
```

Qué grafo estamos consultando?

Grafos nombrados

```
select ?lugarNac ?pais
where
{
    dbr:María_Noel_Riccetto dbo:birthPlace ?lugarNac .
    ?lugarNac dbo:country ?pais .
}
```

Qué grafo estamos consultando?

El **grafo por defecto** de la base de datos RDF: <http://dbpedia.org>

Grafos nombrados

Un **RDF Dataset** es **conjunto de grafos RDF** que tiene:

- un **grafo por defecto**
- cero o más **grafos nombrados** (*named graphs*), cada uno de ellos **identificados por una IRI**.

Grafos nombrados

Un **RDF Dataset** es **conjunto de grafos RDF** que tiene:

- un **grafo por defecto**
- cero o más **grafos nombrados** (*named graphs*), cada uno de ellos **identificados por una IRI**.

Endpoint DBPEDIA:

- El grafo por defecto <http://dbpedia.org> incluye todo el vocabulario
- Grafos nombrados identificados por **IRIs**:
<http://www.w3.org/2002/07/owl#>, http://dbpedia.org/schema/property_rules#,
<http://dbpedia.org/resource/classes#>,

Grafos nombrados

Conjunto de recursos que son instancias de las clases del grafo nombrado <http://dbpedia.org/resource/classes#>.

```
select ?recurso ?nombre ?clase
FROM named <http://dbpedia.org/resource/classes#>
where
{
  ?recurso rdf:type ?clase .
  ?recurso rdfs:label ?nombre .
  {
    graph <http://dbpedia.org/resource/classes#>
    {?clase rdf:type owl:Class .}
  }
}
```

Grafos nombrados

Conjunto de recursos que son instancias de las clases del grafo nombrado <http://dbpedia.org/resource/classes#>.

```
select ?recurso ?nombre ?clase
```

```
FROM named <http://dbpedia.org/resource/classes#>  Grafo nombrado
```

```
where
```

```
{
```

```
?recurso rdf:type ?clase .  Evaluación en grafo por defecto
```

```
?recurso rdfs:label ?nombre .
```

```
{
```

```
graph <http://dbpedia.org/resource/classes#>
```

```
{?clase rdf:type owl:Class .}
```

```
 Evaluación en grafo nombrado
```

```
}
```

```
}
```

Grafos nombrados

Conjunto de recursos que son instancias de las clases del grafo nombrado <http://dbpedia.org/resource/classes#>.

recurso	nombre	clase
http://dbpedia.org/resource/OpenLink_Software	"OpenLink Software"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/OpenLink_Software	"OpenLink Software"@nl	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015 شركة"@ar	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015, Inc."@ca	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015 Games"@de	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015 Games, LLC."@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015 Inc"@fr	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015 (企業)"@ja	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015 (기업)"@ko	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015, Inc."@nl	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015, Inc."@pt	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games_LLC	"2015, Inc."@ru	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5_Rue_Christine	"5 Rue Christine"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5_Rue_Christine	"5 Rue Christine"@nl	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5th_Cell	"5th Cell"@de	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5th_Cell	"5th Cell"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5th_Cell	"5th Cell"@fr	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/606_Films	"606 Films"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/6126_(clothing_line)	"6126 (clothing line)"@en	http://dbpedia.org/ontology/Company

Grafos nombrados

Conjunto de recursos que son instancias de las clases del grafo nombrado <http://dbpedia.org/resource/classes#>.

recurso	nombre	clase
http://dbpedia.org/resource/OpenLink_Software	"OpenLink Software"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/OpenLink_Software	"OpenLink Software"@nl	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015 شركة"@ar	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015, Inc."@ca	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015 Games"@de	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015 Games, LLC."@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015 Inc"@fr	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015 (企業)"@ja	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015 (기업)"@ko	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015, Inc."@nl	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015, Inc."@pt	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/2015_Games,_LLC	"2015, Inc."@ru	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5_Rue_Christine	"5 Rue Christine"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5_Rue_Christine	"5 Rue Christine"@nl	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5th_Cell	"5th Cell"@de	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5th_Cell	"5th Cell"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/5th_Cell	"5th Cell"@fr	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/606_Films	"606 Films"@en	http://dbpedia.org/ontology/Company
http://dbpedia.org/resource/6126_(clothing_line)	"6126 (clothing line)"@en	http://dbpedia.org/ontology/Company

Grafos nombrados – cambiar el grafo por defecto

Propiedades del grafo <http://dbpedia.org/resource/classes#> que tienen propiedades equivalentes en el grafo http://dbpedia.org/schema/property_rules#, su nombre y su dominio.

```
select ?prop ?nombre ?dominio ?propEquiv
FROM <http://dbpedia.org/resource/classes#>
FROM named <http://dbpedia.org/schema/property_rules#>
where
{
    ?prop rdf:type rdf:Property .
    ?prop rdfs:label ?nombre .
    ?prop rdfs:domain ?dominio .
    graph <http://dbpedia.org/schema/property_rules#>
    {
        ?prop owl:equivalentProperty ?propEquiv .
    }
}
```

Grafos nombrados – cambiar el grafo por defecto

Propiedades del grafo <http://dbpedia.org/resource/classes#> que tienen propiedades equivalentes en el grafo http://dbpedia.org/schema/property_rules#, su nombre y su dominio.

```
select ?prop ?nombre ?dominio ?propEquiv
```

```
FROM <http://dbpedia.org/resource/classes#>  Grafo por defecto
```

```
FROM named <http://dbpedia.org/schema/property_rules#>
```

```
where
```

```
{
```

```
  ?prop rdf:type rdf:Property .
```

```
  ?prop rdfs:label ?nombre .
```

```
  ?prop rdfs:domain ?dominio .
```

```
graph <http://dbpedia.org/schema/property_rules#>
```

```
{
```

```
  ?prop owl:equivalentProperty ?propEquiv .
```

```
}
```

```
}
```


Grafos nombrados – cambiar el grafo por defecto

Propiedades del grafo <http://dbpedia.org/resource/classes#> que tienen propiedades equivalentes en el grafo http://dbpedia.org/schema/property_rules#, su nombre y su dominio.

prop	nombre	dominio	propEquiv
http://dbpedia.org/ontology/deathPlace	"death place"@en	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"death place"@en	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"Sterbeort"@de	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"Sterbeort"@de	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"lieu de décès"@fr	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"lieu de décès"@fr	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"plaats van overlijden"@nl	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"plaats van overlijden"@nl	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"τόπος_θανάτου"@el	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"τόπος_θανάτου"@el	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"死没地"@ja	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"死没地"@ja	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"death place"@en	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"death place"@en	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace
http://dbpedia.org/ontology/deathPlace	"Sterbeort"@de	http://dbpedia.org/ontology/Person	http://dbpedia.org/property/deathPlace

Más de un grafo nombrado, variables


Propiedades del grafo <http://dbpedia.org/resource/classes#> que tienen propiedades equivalentes en el grafo http://dbpedia.org/schema/property_rules#, y su dominio.

```
select ?prop ?dominio ?propEquiv
FROM named <http://dbpedia.org/resource/classes#>
FROM named <http://dbpedia.org/schema/property_rules#>
where
{
  graph ?g1
  {
    ?prop rdf:type rdf:Property .
    ?prop rdfs:domain ?dominio .
  }
  graph ?g2
  {
    ?prop owl:equivalentProperty ?propEquiv .
  }
}
```

Las condiciones se evalúan en cada uno de los grafos nombrados

Consultas federadas - SERVICE

Para cada jugador obtener equipo, la profesión y si conoce a otro jugador, en endpoint GraphDB



Consultas federadas - SERVICE

Para cada jugador obtener equipo, la profesión y si conoce a otro jugador, en endpoint GraphDB

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
PREFIX schema: <http://schema.org/>
```

```
select ?jugador ?equipo ?profesion ?conocido
```


```
where {
```

```
    ?jugador dbo:team ?equipo .
```

```
    ?jugador schema:jobTitle ?profesion .
```

```
    OPTIONAL{?jugador schema:knows ?conocido .}
```

```
}
```



Consultas federadas - SERVICE

Para cada jugador obtener equipo, la profesión y si conoce a otro jugador, en endpoint GraphDB, y **agregar el lugar de nacimiento** (DBPEDIA)

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
PREFIX schema: <http://schema.org/>
```

```
select ?jugador ?equipo ?profesion ?lugarNac ?conocido
```

```
where {
```


```
    ?jugador dbo:team ?equipo .
```

```
    ?jugador schema:jobTitle ?profesion .
```

```
    OPTIONAL{?jugador schema:knows ?conocido .}
```

```
    OPTIONAL{?jugador dbo:birthPlace ?lugarNac .}
```

```
}
```



Consultas federadas - SERVICE

Para cada jugador obtener equipo, la profesión y si conoce a otro jugador, en endpoint GraphDB, y **agregar el lugar de nacimiento (DBPEDIA)**

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
PREFIX schema: <http://schema.org/>
```

```
select ?jugador ?equipo ?profesion ?lugarNac ?conocido
```

```
where {
```

```
    ?jugador dbo:team ?equipo .
```

```
    ?jugador schema:jobTitle ?profesion .
```

```
    OPTIONAL{?jugador schema:knows ?conocido .
```

```
    OPTIONAL{?jugador dbo:birthPlace ?lugarNac .}
```

```
}
```

jugador	equipo	profesion	lugarNac	conocido
dbr:Luis_Suárez	dbr:FC_Barcelona	dbr:Footballer		
dbr:Diego_Forlán	dbr:Peñarol	dbr:Footballer		dbr:Luis_Suárez
dbr:Diego_Forlán	dbr:Peñarol	dbr:Coach		dbr:Luis_Suárez

Consultas federadas - SERVICE

Para cada jugador obtener equipo, la profesión y si conoce a otro jugador, en endpoint GraphDB, y **agregar el lugar de nacimiento (DBPEDIA)**

```
PREFIX dbo: <http://dbpedia.org/ontology/>
PREFIX schema: <http://schema.org/>

select ?jugador ?equipo ?profesion ?lugarNac ?conocido
where {
    ?jugador dbo:team ?equipo .
    ?jugador schema:jobTitle ?profesion .
    OPTIONAL{?jugador schema:knows ?conocido .}
    SERVICE <https://dbpedia.org/sparql>
        {?jugador dbo:birthPlace ?lugarNac . }
}
```

Consultas federadas - SERVICE

Para cada jugador obtener equipo, la profesión y si conoce a otro jugador, en endpoint GraphDB, y **agregar el lugar de nacimiento (DBPEDIA)**

```
PREFIX dbo: <http://dbpedia.org/ontology/>
```

```
PREFIX schema: <http://schema.org/>
```

```
select ?jugador ?equipo ?profesion ?conocido
```

```
where {
```

```
  ?jugador dbo:team ?equipo .
```

```
  ?jugador schema:jobTitle ?profesion .
```

```
  OPTIONAL{?jugador schema:knows ?conocido .}
```

```
SERVICE <https://dbpedia.org/sparql>  Endpoint remoto
```

```
  {?jugador dbo:birthPlace ?lugarNac . }
```


```
}
```

jugador	equipo	profesion	lugarNac	conocido
dbr:Diego_Forlán	dbr:Peñarol	dbr:Coach	dbr:Montevideo	dbr:Luis_Suárez
dbr:Diego_Forlán	dbr:Peñarol	dbr:Footballer	dbr:Montevideo	dbr:Luis_Suárez
dbr:Luis_Suárez	dbr:FC_Barcelona	dbr:Footballer	dbr:Salto,_Uruguay	

Consultas federadas - SERVICE

El constructor **SERVICE** hace que el motor SPARQL **procese una porción de la consulta contra un endpoint remoto.**

Los resultados parciales de cada porción de la consulta se integran para devolver el resultado de la consulta.



Axel-Cyrille Ngonga Ngomo, Muhammad Saleer
Federated Query Processing: Challenges and Opportunities. PROFILES@ESWC 2016

RDF(S) - SPARQL – Bibliografía

- SPARQL 1.1 Query Language <https://www.w3.org/TR/sparql11-query/>
- SPARQL 1.1 Federated Query - W3C Recommendation 21 March 2013. <https://www.w3.org/TR/2013/REC-sparql11-federated-query-20130321/>
- Jorge Pérez, Marcelo Arenas, Claudio Gutierrez. Semantics and Complexity of SPARQL. 5th International Semantic Web Conference, ISWC 2006, Athens, GA, USA, November 5-9, 2006.
- P. Hitzler, S. Rudolph, M. Krötzsch: Foundations of Semantic Web Technologies. CRC Press, 2009. (Chapter 7)
- Dbpedia. <https://www.dbpedia.org/about/>